REVIEW OF SERIES EXPANSION

Introduction

In the second half of the course, we will focus quite a bit of attention on the use of series expansions in physics and mathematics. We will spend several weeks studying Fourier series (Ch. 7 in Boas) , series solutions of differential equations (Ch. 12 of Boas) as well as Legendre series (also Ch. 12 Boas).

Since we will be spending so much time studying series and series solutions, it is wise to review what we know, and develop some general principles which can focus our understanding of series. Additionally, this write up will show how we can use Mathematica to support and inform our study of series.

Reviewing Taylor Series

In first year calculus, you undoubtedly spent significant time studying Taylor series. The general idea behind Taylor series is that if a function satisfies certain criteria , then you can express the function as an infinite series of polynomials. In its most general terms, the value of a function, f (x), in the vicinity of the point $x_0 = a$, is given by :

$$f(x) = f(a) + \frac{f'(a)(x-a)}{1!} + \frac{f''(a)(x-a)^2}{2!} + \frac{f'''(a)(x-a)^3}{3!} + \ldots$$

which can be expressed in the more economical summation form :

$$f(x) = \frac{\sum_{n=0}^{\infty} f^{(n)}(a)(x-a)^n}{n!} \qquad (1)$$

where $f^{(n)}(a)$ is the n-th derivative of the function evaluated at the point x_0=a; and n! means n factorial.

Let's see what this equation means by using it to determine the value of $e^{2.1}$. In order to use equation (1), we will evaluate the function f(x) = e^x in the vicinity of the point a=2.0. Obviously, the terms $(x-a)^n$ will be $(2.1-2.0)^n = 0.1^n$.

All the derivatives of e^x are e^x , so equation (1) becomes:

$$f(2.1) = f(2) + \frac{f'(2)(0.1)^1}{1} + \frac{f''(2)(0.1)^2}{2!} + \frac{f'''(2)(0.1)^3}{3!} + \ldots \Rightarrow$$

$$f(2.1) = e^2 + e^2(0.1) + \frac{e^2(0.01)}{2} + \frac{e^2(0.001)}{6} + \ldots$$

$$e^2[1 + 0.1 + 0.005 + 0.00016 + \ldots] = 8.1661$$

Based on only the first four terms of the Taylor expansion of e^x in the vicinity of x_0=2, we approximate the value of $e^{2.1}$ as 8.1661. Using *Mathematica* as comparison, we see:

```
In[8]:= Exp[2.1]
Out[8]= 8.16617
```

that this was not a bad approximation at all.

A special case of the Taylor series is the Maclaurin series, in which you use this technique to determine the value of a function in the vicinity of the point $x_0 = 0$. This leads to the obvious simplification of equation (1):

$$f(x) = \sum_0^{\infty} \frac{f^{(n)}(0)\, x^n}{n!}$$

Thus, to find the Maclaurin expansion of any function (that is suitably behaved), you need to take successive derivatives of the function, and then evaluate those derivatives at x_0= a =0. A well known Maclaurin function, and one you probably did explicitly in Calc I or II, is the expansion for the function:

$$f(x) = \frac{1}{1-x}$$

When evaluated at x = 0, f (0) = 1. Let' s look at successive derivatives of the function :

$$f'(x) = \frac{1}{(1-x)^2} \Rightarrow f'(0) = 1$$

$$f''(x) = \frac{2}{(1-x)^3} \Rightarrow f''(0) = 2$$

$$f'''(x) = \frac{6}{(1-x)^4} \Rightarrow f'''(0) = 6$$

or in general form, the nth derivative is :

$$f^{(n)}(x) = \frac{n!}{(1-x)^{n+1}} \Rightarrow f^{(n)}(0) = n!$$

This particularly simple form allows us to write the Maclaurin expansion very quickly :

$$f(x) = \frac{1}{1-x} = \sum_{n=0}^{\infty} \frac{n!\, x^n}{n!} = \sum_{n=0}^{\infty} x^n = 1 + x + x^2 + x^3 + x^4 + \ldots$$

However, this expression is meaningful only as long as $|x| < 1$, since it is clear that the denominator goes to zero as x approaches 1 (recall your discussions of radius of convergence for series).

Looking at Taylor Series with Greater Depth :

Ok, you have seen this all before, but do you really believe it? Do you really believe that the function f(x) = 1/(1 - x), or any other function like cos x or arc tan x can be accurately approximated by an infinite sum of powers of x? Let' s make use of Mathematica to see how well the series expansion represents f(x) = 1/(1 - x).

Let's see what happens if we compare the plots of f(x)=1/(1-x) with the plots of the series expansion on the interval (0,1). First, let's see what happens if we compare 1/(1-x) with the first two terms of the expansion. In other words, let's plot on the same graph 1/(1-x) and (1+x):

In[15]:=

```
Plot[{1 / (1 - x), 1 + x}, {x, 0, 1}]
```

Out[15]=



Interesting. The two curves are pretty close until about x = 0.15, after which they noticeably diverge. Ok, let' s plot 1/(1 - x) and the first four terms of the series expansion :

In[16]:=

```
Plot[{1 / (1 - x), 1 + x + x^2 + x^3}, {x, 0, 1}]
```

Out[16]=



Even more interesting. the two curves are very close now until about x = 0.4, and the divergence between them is less than before, so we have some confidence that adding more terms might bring the two curves into closer and closer alignment. But, we can also see that it will get pretty boring pretty quickly if we have to keep writing explicit expansion sequences in our Plot statements.

Can we find a more efficient way to do this in Mathematica? Of course we can ... let' s take a short Mathematica interlude and learn how to do sums in Mathematica.

Mathematica Interlude I: Taking Sums

Summations in Mathematica are done using the Sum command. Suppose we want to sum the first 21 terms in the series expansion :

$$f(x) = \frac{1}{1-x} = \sum_{n=0}^{\infty} x^n$$

To instruct Mathematica to sum the first 21 terms of this series, we write :

```
Sum[x^n, {n, 0, 20}]
```

(Remember, since we are starting at n=0, we are summing over 21 terms culminating with the x^{20} term). The command, Sum, is capitalized and uses square brackets. The first element inside the brackets is the expression we are summing over, in this case x^n. The information inside the braces tells you (in order): the index over which you are summing (here that is *n*), the starting value of *n* (*n*=0) and the final value (in this case, *n* =20.

So, let' s see how well this series converges to our function, f (x) = 1/(1 - x). If x = 1/2, it is trivial to calculate that f (1/2) = 2; let' s see what we obtain if we calculate the sum of the first 21 terms of this expansion :

```
In[20]:= Clear[x]
         x = 1 / 2;
         Sum[x^n, {n, 0, 20}] // N
Out[22]= 2.
```

The notation : // N instructs Mathematica to return a number in decimal form, without this instruction, we get :

```
In[23]:= Clear[x]
         x = 1 / 2;
         Sum[x^n, {n, 0, 20}]
```

Out[25]= $\frac{2\,097\,151}{1\,048\,576}$

which you can see is pretty close to the value of 2.

To see another example of how the summation function works, let's investigate the function e^x. We know the Maclaurin expansion for $e^x$is:

$$e^x = \sum_{n=0}^{\infty} \frac{x^n}{n!}$$

Let' s see how well this series expansion approximates the value of the exponential function for x = 100. We can use Mathematica to compute :

In[27]:=
```
Exp[100] // N
```

Out[27]= 2.68812×10^{43}

Ok, this is a pretty big number. We might need quite a few terms in the expansion to approximate this. Let' s start with the first 21 terms of the expansion :

In[31]:=
```
Clear[x]
x = 100;
Sum[x^n / n!, {n, 0, 20}] // N
```

Out[33]= 5.12237×10^{21}

Not such a good match. We should not be too surprised since we curtailed our expansion at n = 20, and the value of the n = 20 term in the expansion is :

In[35]:= $\frac{x^n}{n!}$ /. n → 20 // N

Out[35]= 4.11032×10^{21}

(Remember these commands; "/.n->20" means we will replace n with 20; // N is the instruction to compute a decimal.)

Ok, let' s try this again; let' s try this expansion over the first 31 terms :

In[39]:=
```
Clear[x]
x = 100;
Sum[x^n / n!, {n, 0, 30}] // N
```

Out[41]= 5.35416×10^{27}

Trying again :

In[60]:=
```
Clear[x]
x = 100;
Sum[x^n / n!, {n, 0, 120}] // N
```

Out[62]= 2.62718×10^{43}

And if we use 121 terms, we come within about 3% of being accurate. But wait, what happens if we keep adding terms, we want to make sure that we don' t overshoot the value of Exp[100]. So let' s see what happens if we try this :

In[72]:=
```
Clear[a, x]
x = 100;
Sum[x^n / n!, {n, 0, a}] /. a → {120, 160, 200} // N
```

Out[74]= $\{2.62718 \times 10^{43}, 2.68812 \times 10^{43}, 2.688117141816010 \times 10^{43}\}$

In this last summation, we treated the upper limit (the number of terms we would include in the summation) as a variable denoted by a, and then used the "/." command to evaluate the sum for three different values of the upper limit. The results above show you the value of the summation using 121, 161 and 201 terms respectively, and that we could truly add an infinite number of terms without changing the value of the sum. This occurs because at sufficiently large values of n, the value of x^n/n! decreases so that successive individual terms do not cause the expansion to diverge.

Back to Series Expansions

Let' s consider again our function 1/(1 - x) on the interval (0, 1). We can see the approach to convergence by treating the number of terms in the expansion as a variable :

In[88]:=
```
Clear[x, a, n]
Plot[{1 / (1 - x), Sum[x^n, {n, 0, a}] /. a → {1, 5, 10}}, {x, 0, 1}]
```

Out[89]=



In this code, the variable a represents the largest value of n over which we sum; the use of "/.a->{1,5,10}" instructs the program to compute sums of x^n summing respectively over the first 2, 6 and 11 terms. Finally, all four plots (including 1/(1-x) are placed on one graph. You can see how summing over more and more terms causes the series to converge to the function.

Mathematica Interlude II

Now we are in a position to see one of the most striking features of Mathematica, the Manipulate command. This is an interactive command, so to fully appreciate this, you will need to use this code in an open session of Mathematica. First, use the following code (as with all *Mathematica* code, be careful to have all the brackets, braces and parentheses in correct order):

In[90]:=
```
Clear[x, n, a]
Manipulate[
 Plot[{1 / (1 - x), Sum[x^n, {n, 0, a}]}, {x, 0, 1}], {a, 1, 100}]
```

Out[91]=



Your output will look like the box above. Notice the slide bar; this bar will allow you to change the number of terms included in the expansion as you slide the bar from its lowest value to its highest value (1 - 100). As you move the slide bar, you will see the lower curve (the one representing the summation of x^n) approach the curve of 1/(1-x) until the two curves are identical. This is powerful visual verification that summation of sufficient terms in the series expansion will cause the expansion to converge to the function.

Some Slightly More Complicated Functions

In choosing functions like 1/(1 - x) and even e^x, we are working with relatively simple functions in that they are both monotonically increasing over their convergence intervals. Let's think about the nature of the Maclaurin expansions for trig functions like sin x and cos x. The plot below shows the well known curves for sin x and cos x for the interval (0,6π):

In[94]:=
```
Plot[{Sin[x], Cos[x]}, {x, 0, 6 π}]
```

Out[94]=


Let' s think about the oscillatory behavior of trig functions in terms of power series expansions; if we believe we can write sin x and cos x as an infinite sum of terms, we realize very quickly that some of the terms must be positive and some must be negative. If we had only positive terms, the function would continually increase; in order to have a function oscillate between the values of + 1 and - 1, successive terms must cause the function to increase or decrease by just the needed amount. Let' s see how quickly the Maclaurin series for cos x will converge. To refresh your memory, the Maclaurin series for cos x is :

$$f(x) = \cos x = 1 - \frac{x^2}{2!} + \frac{x^4}{4!} - \frac{x^6}{6!} + \ldots \frac{(-1)^n x^{2n}}{(2n)!} \quad (2)$$

As shown in class, we can use the Manipulate feature to help us see the approach to convergence of the series expansion for cos x to the function. The commands below instruct *Mathematica* to construct two plots over the interval (0,6π) on the same graph. The two functions plotted are cos x and the partial sum of the series expansion described in eq. (2) above. In the Sum command, the first part is the expression for the n-th term in the series; the second element of the Sum command is the expression in braces, {n,0,a} which means to sum over the index n, starting from n=0 and ending when n=*a*. As you have seen before, we let *a* be a floating variable, so we can see what the graph of the series expansion looks like as we add more and more terms to the partial sum. The final set of braces, {a, 0, 30} allows us to determine the upper limit of summation. The expression {a,0,30} means to set the variable *a* initially to zero and vary the value of *a* from 0 to 30 as you move the slide bar. When *a*=1, we are summing over the first two terms of the series (the zeroth term and the first term of the expansion). You should see nicely how the series expansion converges to the function f(x)=cos x as you add more and more terms to the expansion.

In[131]:=

```
Clear[x, n, a]
Manipulate[Plot[{Cos[x], Sum[(-1)^n (x^(2 n)) / (2 n) !, {n, 0, a}]},
  {x, 0, 6 π}], {a, 0, 30}]
```

Out[132]=



Some Familar Series

There are some Taylor and Maclaurin series that are so well known and so frequently used, that it is wise to know them and their properties. These include :

$$\frac{1}{1-x} = 1 + x + x^2 + x^3 + x^4 + \ldots x^n = \sum_{n=0}^{\infty} x^n \text{ for } \left| x \right| < 1$$

$$e^x = 1 + x + \frac{x^2}{2!} + \frac{x^3}{3!} + \ldots \frac{x^n}{n!} = \sum_{n=0}^{\infty} \frac{x^n}{n!} \text{ for all x}$$

$$\cos x = 1 - \frac{x^2}{2!} + \frac{x^4}{4!} - \frac{x^6}{6!} + \frac{x^8}{8!} + \ldots \frac{(-1)^n x^{2n}}{(2n)!} = \sum_{n=0}^{\infty} \frac{(-1)\wedge n\, x^{2n}}{(2n)!} \text{ for all x}$$

$$\sin x = x - \frac{x^3}{3!} + \frac{x^5}{5!} - \frac{x^7}{7!} + \ldots \frac{(-1)^n x^{2n+1}}{(2n+1)!} = \frac{\Sigma_{n=0}^{\infty} (-1)^n x^{2n+1}}{(2n+1)!} \text{ for all x}$$

$$\sqrt{1-x} = 1 - \frac{x}{2} - \frac{x^2}{8} - \frac{x^3}{16} - \frac{5x^4}{128} - \ldots$$

Manipulating Series Expansions

One of the most powerful aspects of series expansions is that simple substitutions or manipulations of a well known series allows you to determine the series expansion for a more complex function without the need for doing laborious multiple differentiations. Let' s start with our well known expansion for 1/(1 - x), and show how it allows us to simply determine the series expansion for many different series, including the series for arc tan x. We begin by writing :

$$\frac{1}{1-x} = 1 + x + x^2 + x^3 + \ldots \tag{3}$$

Let' s say we now want to know the series expansion for a similar series, 1/(1 + x). We could use the formula for Taylor series and compute the value of the n - th derivative at x = 0, or we can look at equation (3) and realize that if we simply substitute (-x) for x on both sides of the equation, we obtain :

$$\begin{gathered} \frac{1}{1-(-x))} = 1 + (-x) + (-x)^2 + (-x)^3 + \ldots \Rightarrow \\ \frac{1}{1+x} = 1 - x + x^2 - x^3 + x^4 - \ldots \end{gathered} \tag{4}$$

We can continue working with eq. (4), substituting x^2 for x on both sides of the equation, and obtain:

$$\frac{1}{1+x^2} = 1 - x^2 + x^4 - x^6 + x^8 - ... \qquad (5)$$

You may remember that $1/(1+x^2)$ is the derivative of arc tan x, so, if we multiply both sides of eq. (5) by dx and then integrate each term, we get:

$$\int \frac{1}{1+x^2}\,dx = \int (1 - x^2 + x^4 - x^6 + ...)\,dx =$$

$$\text{Arc tan x} = x - \frac{x^3}{3} + \frac{x^5}{5} - \frac{x^7}{7} + ... \qquad (6)$$

(What happened to the constant of integration that we know we get when we do an indefinite integral? How can you prove that the constant in this case equals zero?)

As a sidenote, let' s see what happens if we set x = 1 in equation (6) above. We know that arctan (1) =π/4; so if set x=1 in equation (6), we get:

$$\text{Arc Tan } 1 = \frac{\pi}{4} = 1 - \frac{1}{3} + \frac{1}{5} - \frac{1}{7} + ...$$

And we derive a simple (but slowly converging) series that allows us to calculate the value of π.

There are many other examples of similar series operations; suppose we want to find the series expansion of of $1/(1-x)^2$. We recognize this is simply the derivative of 1/(1-x), so we can take equation (3), and differentiate term by term to obtain:

$$\frac{1}{(1-x)^2} = 1 + 2x + 3x^2 + ... \qquad (7)$$

Calculations Involving Taylor Series

Suppose we want to find the series expansion of a more complicated function like :

$$\frac{e^x}{\cos x}$$

in the vicinity of x = 0. Since cos x does not vanish at x = 0, and because both e^x and cos x have well determined series expansions, we can be hopeful that $e^x/\cos x$ will also have a well defined series expansion. That means we anticipate that we can construct a series such that:

$$f(x) = \frac{e^x}{\cos x} = \sum_{n=0}^{\infty} a_n x^n = a_0 + a_1 x + a_2 x^2 + a_3 x^3 + ... \tag{8}$$

In short, we know exactly the form of this series expansion; all we have to do is determine the value of the various coefficients. We know one method that will work; we can take successive derivatives of $e^x / \cos x$ and evaluate those derivatives at 0. However, our experience with differentiating quotients should give us some pause, since we know that each differentiation produces a more and more complex term.

There is another technique we can use. Since we already have the form of the series expansion for this f(x), suppose we multiply each side by cos x in eq. (8) to produce:

$$e^x = \left(a_0 + a_1 x + a_2 x^2 + a_3 x^3 + ...\right) \cos x \tag{9}$$

Using the known series expansions for e^x and cos x, eq. (9) becomes:

$$\begin{gathered} 1 + x + \frac{x^2}{2} + \frac{x^3}{6} + ... = \\ \left(a_0 + a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4 ...\right)\left(1 - \frac{x^2}{2} + \frac{x^4}{24} - ...\right) \end{gathered} \tag{10}$$

We can multiply out the right hand side of (10) to get :

$$\begin{gathered} 1 + x + \frac{x^2}{2} + \frac{x^3}{6} + ... = \\ a_0 - a_0 \frac{x^2}{2} + a_0 \frac{x^4}{24} + a_1 x - a_1 \frac{x^3}{2} + a_1 \frac{x^5}{24} + a_2 x^2 - \\ a_2 \frac{x^4}{2} + a_2 \frac{x^6}{24} + a_3 x^3 - a_3 \frac{x^5}{2} + a_4 x^4 - a_4 \frac{x^6}{2} + ...\Big) \end{gathered} \tag{11}$$

If we group the terms on the right side of eq. (11) according to powers of x, we obtain :

$$1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + ... = \\ a_0 + a_1\,x + \left(a_2 - \frac{a_0}{2}\right)x^2 + \left(a_3 - \frac{a_1}{2}\right)x^3 + \left(\frac{a_0}{24} - \frac{a_2}{2} + a_4\right)x^4 \tag{12}$$

In order for the two sides of eq. (12) to be equal, we know that the coefficient of the x^n term on the left must equal the coeffiecient of the similar term on the right; thus, by inspecting each side of eq. (12), we conclude that:

$$\begin{aligned} a_0 &= 1 \\ a_1 &= 1 \\ a_2 - \frac{a_0}{2} = \frac{1}{2} &\Rightarrow a_2 = \frac{a_0}{2} + \frac{1}{2} = 1 \\ a_3 - \frac{a_1}{2} = \frac{1}{6} &\Rightarrow a_3 = \frac{a_1}{2} + \frac{1}{6} = \frac{2}{3} \\ a_4 - \frac{a_2}{2} + \frac{a_0}{24} = \frac{1}{24} &\Rightarrow a_4 = \frac{a_2}{2} - \frac{a_0}{24} + \frac{1}{24} = \frac{1}{2} \end{aligned} \tag{13}$$

Having worked out these coefficients, we can now write that the Taylor Series for our function is :

$$f\,(x) = \frac{e^x}{\cos x} = 1 + x + x^2 + \frac{2\,x^3}{3} + \frac{x^4}{2} + ... \tag{14}$$

Mathematica Interlude III

The Mathematica command Series outputs the series expansion of functions. To get started, let' s see how we use Mathematica to get the first few terms in the series expansion for cos x. We input :

In[151]:=

```
Series[Cos[x], {x, 0, 10}]
```

Out[151]= $1 - \frac{x^2}{2} + \frac{x^4}{24} - \frac{x^6}{720} + \frac{x^8}{40\,320} - \frac{x^{10}}{3\,628\,800} + O[x]^{11}$

The first element in the Series call is obviously the function we wish to expand; the braces {x, 0, 10} instruct Mathematica to expand this function in the neighborhood of x = 0, and write out the series up to and including the x^{10} term. The last term in the output, $O[x]^{11}$ is the mathematical way of writing "plus terms of order x^{11} and higher".

We can produce output omitting the "O[x]" term if we input :

In[152]:=

```
Series[Cos[x], {x, 0, 10}] // Normal
```

Out[152]= $1-\frac{x^2}{2}+\frac{x^4}{24}-\frac{x^6}{720}+\frac{x^8}{40320}-\frac{x^{10}}{3628800}$

Verifying our series expansion for f (x) $= e^x / \cos x$:

In[155]:=

```
Series[Exp[x] / Cos[x], {x, 0, 4}]
```

Out[155]= $1+x+x^2+\frac{2x^3}{3}+\frac{x^4}{2}+O[x]^5$